# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Andri Yngvason <andri@yngvason.is>

pkgname=aml
pkgver=0.3.0
pkgrel=1
pkgdesc='Another Main Loop'
arch=(loong64 x86_64)
url=https://github.com/any1/aml
license=(custom:ISC)
depends=(glibc)
makedepends=(
  git
  meson
  ninja
)
provides=(libaml.so)
_tag=b83f3576ce4187d9285f06e9066ef43a691464d4
source=(git+https://github.com/any1/aml.git#tag=${_tag})
b2sums=(SKIP)

pkgver() {
  cd aml
  git describe --tags | sed 's/^v//'
}

build() {
  arch-meson aml build
  meson compile -C build
}

package() {
  DESTDIR="${pkgdir}" meson install -C build
  install -Dm 644 aml/COPYING -t "${pkgdir}"/usr/share/licenses/aml
}

# vim: ts=2 sw=2 et:
